



&lt;&gt; Code

Issues 9

Pull requests 1

Actions

Projects

Security



bee9759 ▾

[account-abstraction](#) / [contracts](#) / [core](#) / EntryPoint.sol 

This commit does not belong to any branch on this repository, and may belong to a fork outside of the repository.



drortirosh Merge branch 'develop' into executeUserOp ✓

last month



783 lines (721 loc) · 29.3 KB

Code

Blame

Raw



```
1 // SPDX-License-Identifier: GPL-3.0
2 pragma solidity ^0.8.23;
3 /* solhint-disable avoid-low-level-calls */
4 /* solhint-disable no-inline-assembly */
5
6 import "../interfaces/IAccount.sol";
7 import "../interfaces/IAccountExecute.sol";
8 import "../interfaces/IPaymaster.sol";
9 import "../interfaces/IEntryPoint.sol";
10
11 import "../utils/Exec.sol";
12 import "../StakeManager.sol";
13 import "../SenderCreator.sol";
14 import "../Helpers.sol";
15 import "../NonceManager.sol";
16 import "../UserOperationLib.sol";
17
18 // we also require '@gnosis.pm/safe-contracts' and both libraries have 'IERC165.sol',
19 import "@openzeppelin/contracts/utils/introspection/ERC165.sol" as OpenZeppelin;
20 import "@openzeppelin/contracts/utils/ReentrancyGuard.sol";
21
22 /*
23  * Account-Abstraction (EIP-4337) singleton EntryPoint implementation.
24  * Only one instance required on each chain.
25  */
26 contract EntryPoint is IEntryPoint, StakeManager, NonceManager, ReentrancyGuard, Open
27
28     using UserOperationLib for UserOperation;
29
30     SenderCreator private senderCreator = new SenderCreator();
31
```

```

32 // Marker for inner call revert on out of gas
33 bytes32 private constant INNER_OUT_OF_GAS = hex"deaddead";
34
35 uint256 private constant REVERT_REASON_MAX_LEN = 2048;
36 uint256 private constant PENALTY_PERCENT = 10;
37
38 /**
39  * For simulation purposes, validateUserOp (and validatePaymasterUserOp)
40  * must return this value in case of signature failure, instead of revert.
41  */
42 uint256 public constant SIG_VALIDATION_FAILED = 1;
43
44 /// @inheritdoc OpenZeppelin.IERC165
45 function supportsInterface(bytes4 interfaceId) public view virtual override return
46     // note: solidity "type(IEntryPoint).interfaceId" is without inherited method
47     return interfaceId == (type(IEntryPoint).interfaceId ^ type(IStakeManager).in
48         interfaceId == type(IEntryPoint).interfaceId ||
49         interfaceId == type(IStakeManager).interfaceId ||
50         interfaceId == type(INonceManager).interfaceId ||
51         super.supportsInterface(interfaceId);
52 }
53
54 /**
55  * Compensate the caller's beneficiary address with the collected fees of all Use
56  * @param beneficiary - The address to receive the fees.
57  * @param amount      - Amount to transfer.
58  */
59 function _compensate(address payable beneficiary, uint256 amount) internal {
60     require(beneficiary != address(0), "AA90 invalid beneficiary");
61     (bool success, ) = beneficiary.call{value: amount}("");
62     require(success, "AA91 failed send to beneficiary");
63 }
64
65 /**
66  * Execute a user operation.
67  * @param opIndex      - Index into the opInfo array.
68  * @param userOp       - The userOp to execute.
69  * @param opInfo       - The opInfo filled by validatePrepayment for this userOp.
70  * @return collected   - The total amount this userOp paid.
71  */
72 function _executeUserOp(
73     uint256 opIndex,
74     UserOperation calldata userOp,
75     UserOpInfo memory opInfo
76 )
77 internal
78 returns
79 (uint256 collected) {
80     uint256 preGas = gasleft();
81     bytes memory context = getMemoryBytesFromOffset(opInfo.contextOffset);
82     uint saveFreePtr;
83     assembly {
84         saveFreePtr := mload(0x40)
85     }

```

```

85     }
86     bytes calldata callData = userOp.callData;
87     bytes memory innerCall;
88     bytes4 methodSig;
89     assembly {
90         let len := callData.length
91         if gt(len,3) {
92             methodSig := calldataload(callData.offset)
93         }
94     }
95     if (methodSig == IAccountExecute.executeUserOp.selector) {
96         bytes memory executeUserOp = abi.encodeCall(IAccountExecute.executeUserOp,
97             innerCall = abi.encodeCall(this.innerHandleOp, (executeUserOp, opInfo, co
98     } else
99     {
100         innerCall = abi.encodeCall(this.innerHandleOp, (callData, opInfo, context
101     }
102     bool success;
103     assembly {
104         success := call(gas(), address(), 0, add(innerCall, 0x20), mload(innerCall
105         collected := mload(0)
106         mstore(0x40, saveFreePtr)
107     }
108     if (!success) {
109         bytes32 innerRevertCode;
110         assembly {
111             let len := returndatasize()
112             if eq(32,len) {
113                 returndatacopy(0, 0, 32)
114                 innerRevertCode := mload(0)
115             }
116         }
117         // handleOps was called with gas limit too low. abort entire bundle.
118         if (innerRevertCode == INNER_OUT_OF_GAS) {
119             //report paymaster, since if it is not deliberately caused by the bun
120             // it must be a revert caused by paymaster.
121             revert FailedOp(opIndex, "AA95 out of gas");
122         } else {
123             emit PostOpRevertReason(
124                 opInfo.userOpHash,
125                 opInfo.mUserOp.sender,
126                 opInfo.mUserOp.nonce,
127                 Exec.getReturnData(REVERT_REASON_MAX_LEN)
128             );
129         }
130
131         uint256 actualGas = preGas - gasleft() + opInfo.preOpGas;
132         collected = _postExecution(
133             opIndex,
134             IPaymaster.PostOpMode.postOpReverted,
135             opInfo,
136             context,
137             actualGas
138         );
139     }

```

```

139         }
140     }
141
142     /// @inheritdoc IEntryPoint
143     function handleOps(
144         UserOperation[] calldata ops,
145         address payable beneficiary
146     ) public nonReentrant {
147         uint256 opslen = ops.length;
148         UserOpInfo[] memory opInfos = new UserOpInfo[](opslen);
149
150         unchecked {
151             for (uint256 i = 0; i < opslen; i++) {
152                 UserOpInfo memory opInfo = opInfos[i];
153                 (
154                     uint256 validationData,
155                     uint256 pmValidationData
156                 ) = _validatePrepayment(i, ops[i], opInfo);
157                 _validateAccountAndPaymasterValidationData(
158                     i,
159                     validationData,
160                     pmValidationData,
161                     address(0)
162                 );
163             }
164
165             uint256 collected = 0;
166             emit BeforeExecution();
167
168             for (uint256 i = 0; i < opslen; i++) {
169                 collected += _executeUserOp(i, ops[i], opInfos[i]);
170             }
171
172             _compensate(beneficiary, collected);
173         }
174     }
175
176     /// @inheritdoc IEntryPoint
177     function handleAggregatedOps(
178         UserOpsPerAggregator[] calldata opsPerAggregator,
179         address payable beneficiary
180     ) public nonReentrant {
181
182         uint256 opasLen = opsPerAggregator.length;
183         uint256 totalOps = 0;
184         for (uint256 i = 0; i < opasLen; i++) {
185             UserOpsPerAggregator calldata opa = opsPerAggregator[i];
186             UserOperation[] calldata ops = opa.userOps;
187             IAggregator aggregator = opa.aggregator;
188
189             //address(1) is special marker of "signature error"
190             require(
191                 address(aggregator) != address(1),
192                 "AA96 invalid aggregator"
193             );

```

```

193         };
194
195         if (address(agggregator) != address(0)) {
196             // solhint-disable-next-line no-empty-blocks
197             try aggregator.validateSignatures(ops, opa.signature) {} catch {
198                 revert SignatureValidationFailed(address(agggregator));
199             }
200         }
201
202         totalOps += ops.length;
203     }
204
205     UserOpInfo[] memory opInfos = new UserOpInfo[](totalOps);
206
207     uint256 opIndex = 0;
208     for (uint256 a = 0; a < opasLen; a++) {
209         UserOpsPerAggregator calldata opa = opsPerAggregator[a];
210         UserOperation[] calldata ops = opa.userOps;
211         IAggregator aggregator = opa.agggregator;
212
213         uint256 opslen = ops.length;
214         for (uint256 i = 0; i < opslen; i++) {
215             UserOpInfo memory opInfo = opInfos[opIndex];
216             (
217                 uint256 validationData,
218                 uint256 paymasterValidationData
219             ) = _validatePrepayment(opIndex, ops[i], opInfo);
220             _validateAccountAndPaymasterValidationData(
221                 i,
222                 validationData,
223                 paymasterValidationData,
224                 address(agggregator)
225             );
226             opIndex++;
227         }
228     }
229
230     emit BeforeExecution();
231
232     uint256 collected = 0;
233     opIndex = 0;
234     for (uint256 a = 0; a < opasLen; a++) {
235         UserOpsPerAggregator calldata opa = opsPerAggregator[a];
236         emit SignatureAggregatorChanged(address(opa.agggregator));
237         UserOperation[] calldata ops = opa.userOps;
238         uint256 opslen = ops.length;
239
240         for (uint256 i = 0; i < opslen; i++) {
241             collected += _executeUserOp(opIndex, ops[i], opInfos[opIndex]);
242             opIndex++;
243         }
244     }
245     emit SignatureAggregatorChanged(address(0));
246

```

```

247         _compensate(beneficiary, collected);
248     }
249
250 /**
251  * A memory copy of UserOp static fields only.
252  * Excluding: callData, initCode and signature. Replacing paymasterAndData with p
253  */
254 struct MemoryUserOp {
255     address sender;
256     uint256 nonce;
257     uint256 callGasLimit;
258     uint256 verificationGasLimit;
259     uint256 preVerificationGas;
260     address paymaster;
261     uint256 maxFeePerGas;
262     uint256 maxPriorityFeePerGas;
263 }
264
265 struct UserOpInfo {
266     MemoryUserOp mUserOp;
267     bytes32 userOpHash;
268     uint256 prefund;
269     uint256 contextOffset;
270     uint256 preOpGas;
271 }
272
273 /**
274  * Inner function to handle a UserOperation.
275  * Must be declared "external" to open a call context, but it can only be called
276  * @param callData - The callData to execute.
277  * @param opInfo - The UserOpInfo struct.
278  * @param context - The context bytes.
279  */
280 function innerHandleOp(
281     bytes memory callData,
282     UserOpInfo memory opInfo,
283     bytes calldata context
284 ) external returns (uint256 actualGasCost) {
285     uint256 preGas = gasleft();
286     require(msg.sender == address(this), "AA92 internal call only");
287     MemoryUserOp memory mUserOp = opInfo.mUserOp;
288
289     uint callGasLimit = mUserOp.callGasLimit;
290     unchecked {
291         // handleOps was called with gas limit too low. abort entire bundle.
292         if (
293             gasleft() < callGasLimit + mUserOp.verificationGasLimit + 5000
294         ) {
295             assembly {
296                 mstore(0, INNER_OUT_OF_GAS)
297                 revert(0, 32)
298             }
299         }
300     }

```

```

301
302     IPaymaster.PostOpMode mode = IPaymaster.PostOpMode.opSucceeded;
303     if (callData.length > 0) {
304         bool success = Exec.call(mUserOp.sender, 0, callData, callGasLimit);
305         if (!success) {
306             bytes memory result = Exec.getReturnData(REVERT_REASON_MAX_LEN);
307             if (result.length > 0) {
308                 emit UserOperationRevertReason(
309                     opInfo.userOpHash,
310                     mUserOp.sender,
311                     mUserOp.nonce,
312                     result
313                 );
314             }
315             mode = IPaymaster.PostOpMode.opReverted;
316         }
317     }
318
319     unchecked {
320         uint256 actualGas = preGas - gasleft() + opInfo.preOpGas;
321         // Note: opIndex is ignored (relevant only if mode==postOpReverted, which
322         return _postExecution(0, mode, opInfo, context, actualGas);
323     }
324 }
325
326 /// @inheritdoc IEntryPoint
327 function getUserOpHash(
328     UserOperation calldata userOp
329 ) public view returns (bytes32) {
330     return
331         keccak256(abi.encode(userOp.hash(), address(this), block.chainid));
332 }
333
334 /**
335  * Copy general fields from userOp into the memory opInfo structure.
336  * @param userOp - The user operation.
337  * @param mUserOp - The memory user operation.
338  */
339 function _copyUserOpToMemory(
340     UserOperation calldata userOp,
341     MemoryUserOp memory mUserOp
342 ) internal pure {
343     mUserOp.sender = userOp.sender;
344     mUserOp.nonce = userOp.nonce;
345     mUserOp.callGasLimit = userOp.callGasLimit;
346     mUserOp.verificationGasLimit = userOp.verificationGasLimit;
347     mUserOp.preVerificationGas = userOp.preVerificationGas;
348     mUserOp.maxFeePerGas = userOp.maxFeePerGas;
349     mUserOp.maxPriorityFeePerGas = userOp.maxPriorityFeePerGas;
350     bytes calldata paymasterAndData = userOp.paymasterAndData;
351     if (paymasterAndData.length > 0) {
352         require(
353             paymasterAndData.length >= 20,
354             "AA93 invalid paymasterAndData"

```

```

355         );
356         mUserOp.paymaster = address(bytes20(paymasterAndData[:20]));
357     } else {
358         mUserOp.paymaster = address(0);
359     }
360 }
361
362 /**
363  * Get the required prefunded gas fee amount for an operation.
364  * @param mUserOp - The user operation in memory.
365  */
366 function _getRequiredPrefund(
367     MemoryUserOp memory mUserOp
368 ) internal pure returns (uint256 requiredPrefund) {
369     unchecked {
370         // When using a Paymaster, the verificationGasLimit is used also to as a
371         // Our security model might call postOp eventually twice.
372         uint256 mul = mUserOp.paymaster != address(0) ? 2 : 1;
373         uint256 requiredGas = mUserOp.callGasLimit +
374             mUserOp.verificationGasLimit *
375             mul +
376             mUserOp.preVerificationGas;
377
378         requiredPrefund = requiredGas * mUserOp.maxFeePerGas;
379     }
380 }
381
382 /**
383  * Create sender smart contract account if init code is provided.
384  * @param opIndex - The operation index.
385  * @param opInfo - The operation info.
386  * @param initCode - The init code for the smart contract account.
387  */
388 function _createSenderIfNeeded(
389     uint256 opIndex,
390     UserOpInfo memory opInfo,
391     bytes calldata initCode
392 ) internal {
393     if (initCode.length != 0) {
394         address sender = opInfo.mUserOp.sender;
395         if (sender.code.length != 0)
396             revert FailedOp(opIndex, "AA10 sender already constructed");
397         address sender1 = senderCreator.createSender{
398             gas: opInfo.mUserOp.verificationGasLimit
399         }(initCode);
400         if (sender1 == address(0))
401             revert FailedOp(opIndex, "AA13 initCode failed or OOG");
402         if (sender1 != sender)
403             revert FailedOp(opIndex, "AA14 initCode must return sender");
404         if (sender1.code.length == 0)
405             revert FailedOp(opIndex, "AA15 initCode must create sender");
406         address factory = address(bytes20(initCode[0:20]));
407         emit AccountDeployed(
408             opInfo.userOpHash,

```



```

409         sender,
410         factory,
411         opInfo.mUserOp.paymaster
412     );
413 }
414 }
415
416 /// @inheritdoc IEntryPoint
417 function getSenderAddress(bytes calldata initCode) public {
418     address sender = senderCreator.createSender(initCode);
419     revert SenderAddressResult(sender);
420 }
421
422 /**
423  * Call account.validateUserOp.
424  * Revert (with FailedOp) in case validateUserOp reverts, or account didn't send
425  * Decrement account's deposit if needed.
426  * @param opIndex      - The operation index.
427  * @param op            - The user operation.
428  * @param opInfo        - The operation info.
429  * @param requiredPrefund - The required prefund amount.
430  */
431 function _validateAccountPrepayment(
432     uint256 opIndex,
433     UserOperation calldata op,
434     UserOpInfo memory opInfo,
435     uint256 requiredPrefund
436 )
437     internal
438     returns (
439         uint256 gasUsedByValidateAccountPrepayment,
440         uint256 validationData
441     )
442 {
443     unchecked {
444         uint256 preGas = gasleft();
445         MemoryUserOp memory mUserOp = opInfo.mUserOp;
446         address sender = mUserOp.sender;
447         _createSenderIfNeeded(opIndex, opInfo, op.initCode);
448         address paymaster = mUserOp.paymaster;
449         uint256 missingAccountFunds = 0;
450         if (paymaster == address(0)) {
451             uint256 bal = balanceOf(sender);
452             missingAccountFunds = bal > requiredPrefund
453                 ? 0
454                 : requiredPrefund - bal;
455         }
456         try
457             IAccount(sender).validateUserOp{
458                 gas: mUserOp.verificationGasLimit
459             }(op, opInfo.userOpHash, missingAccountFunds)
460         returns (uint256 _validationData) {
461             validationData = _validationData;
462         } catch {

```

```

463         revert FailedOpWithRevert(opIndex, "AA23 reverted", Exec.getReturnDat
464     }
465     if (paymaster == address(0)) {
466         DepositInfo storage senderInfo = deposits[sender];
467         uint256 deposit = senderInfo.deposit;
468         if (requiredPrefund > deposit) {
469             revert FailedOp(opIndex, "AA21 didn't pay prefund");
470         }
471         senderInfo.deposit = uint112(deposit - requiredPrefund);
472     }
473     gasUsedByValidateAccountPrepayment = preGas - gasleft();
474 }
475 }
476
477 /**
478  * In case the request has a paymaster:
479  * - Validate paymaster has enough deposit.
480  * - Call paymaster.validatePaymasterUserOp.
481  * - Revert with proper FailedOp in case paymaster reverts.
482  * - Decrement paymaster's deposit.
483  * @param opIndex - The operation index.
484  * @param op - The user operation.
485  * @param opInfo - The operation info.
486  * @param requiredPreFund - The required prefund amount.
487  * @param gasUsedByValidateAccountPrepayment - The gas used by _validateAccountPr
488  */
489 function _validatePaymasterPrepayment(
490     uint256 opIndex,
491     UserOperation calldata op,
492     UserOpInfo memory opInfo,
493     uint256 requiredPreFund,
494     uint256 gasUsedByValidateAccountPrepayment
495 ) internal returns (bytes memory context, uint256 validationData) {
496     unchecked {
497         MemoryUserOp memory mUserOp = opInfo.mUserOp;
498         uint256 verificationGasLimit = mUserOp.verificationGasLimit;
499         require(
500             verificationGasLimit > gasUsedByValidateAccountPrepayment,
501             "AA41 too little verificationGas"
502         );
503         uint256 gas = verificationGasLimit -
504             gasUsedByValidateAccountPrepayment;
505
506         address paymaster = mUserOp.paymaster;
507         DepositInfo storage paymasterInfo = deposits[paymaster];
508         uint256 deposit = paymasterInfo.deposit;
509         if (deposit < requiredPreFund) {
510             revert FailedOp(opIndex, "AA31 paymaster deposit too low");
511         }
512         paymasterInfo.deposit = uint112(deposit - requiredPreFund);
513         try
514             IPaymaster(paymaster).validatePaymasterUserOp{gas: gas}(
515                 op,
516                 opInfo.userOpHash,

```

```

517         requiredPreFund
518     )
519     returns (bytes memory _context, uint256 _validationData) {
520         context = _context;
521         validationData = _validationData;
522     } catch {
523         revert FailedOpWithRevert(opIndex, "AA33 reverted", Exec.getReturnData());
524     }
525 }
526 }
527
528 /**
529  * Revert if either account validationData or paymaster validationData is expired
530  * @param opIndex          - The operation index.
531  * @param validationData    - The account validationData.
532  * @param paymasterValidationData - The paymaster validationData.
533  * @param expectedAggregator - The expected aggregator.
534  */
535 function _validateAccountAndPaymasterValidationData(
536     uint256 opIndex,
537     uint256 validationData,
538     uint256 paymasterValidationData,
539     address expectedAggregator
540 ) internal view {
541     (address aggregator, bool outOfTimeRange) = _getValidationData(
542         validationData
543     );
544     if (expectedAggregator != aggregator) {
545         revert FailedOp(opIndex, "AA24 signature error");
546     }
547     if (outOfTimeRange) {
548         revert FailedOp(opIndex, "AA22 expired or not due");
549     }
550     // pmAggregator is not a real signature aggregator: we don't have logic to handle it
551     // Non-zero address means that the paymaster fails due to some signature check
552     address pmAggregator;
553     (pmAggregator, outOfTimeRange) = _getValidationData(
554         paymasterValidationData
555     );
556     if (pmAggregator != address(0)) {
557         revert FailedOp(opIndex, "AA34 signature error");
558     }
559     if (outOfTimeRange) {
560         revert FailedOp(opIndex, "AA32 paymaster expired or not due");
561     }
562 }
563
564 /**
565  * Parse validationData into its components.
566  * @param validationData - The packed validation data (sigFailed, validAfter, validBefore)
567  */
568 function _getValidationData(
569     uint256 validationData
570 ) internal view returns (address aggregator, bool outOfTimeRange) {

```

```

571         if (validationData == 0) {
572             return (address(0), false);
573         }
574         ValidationData memory data = _parseValidationData(validationData);
575         // solhint-disable-next-line not-rely-on-time
576         outOfTimeRange = block.timestamp > data.validUntil || block.timestamp < data.
577         aggregator = data.aggregator;
578     }
579
580     /**
581      * Validate account and paymaster (if defined) and
582      * also make sure total validation doesn't exceed verificationGasLimit.
583      * This method is called off-chain (simulateValidation()) and on-chain (from hand
584      * @param opIndex - The index of this userOp into the "opInfos" array.
585      * @param userOp - The userOp to validate.
586      */
587     function _validatePrepayment(
588         uint256 opIndex,
589         UserOperation calldata userOp,
590         UserOpInfo memory outOpInfo
591     )
592         internal
593         returns (uint256 validationData, uint256 paymasterValidationData)
594     {
595         uint256 preGas = gasleft();
596         MemoryUserOp memory mUserOp = outOpInfo.mUserOp;
597         _copyUserOpToMemory(userOp, mUserOp);
598         outOpInfo.userOpHash = getUserOpHash(userOp);
599
600         // Validate all numeric values in userOp are well below 128 bit, so they can
601         // and multiplied without causing overflow.
602         uint256 maxGasValues = mUserOp.preVerificationGas |
603             mUserOp.verificationGasLimit |
604             mUserOp.callGasLimit |
605             userOp.maxFeePerGas |
606             userOp.maxPriorityFeePerGas;
607         require(maxGasValues <= type(uint120).max, "AA94 gas values overflow");
608
609         uint256 gasUsedByValidateAccountPrepayment;
610         uint256 requiredPreFund = _getRequiredPrefund(mUserOp);
611         (
612             gasUsedByValidateAccountPrepayment,
613             validationData
614         ) = _validateAccountPrepayment(
615             opIndex,
616             userOp,
617             outOpInfo,
618             requiredPreFund
619         );
620
621         if (!_validateAndUpdateNonce(mUserOp.sender, mUserOp.nonce)) {
622             revert FailedOp(opIndex, "AA25 invalid account nonce");
623         }
624

```



```

670         try IPaymaster(paymaster).postOp(
671             gas: mUserOp.verificationGasLimit
672         )(mode, context, actualGasCost)
673         // solhint-disable-next-line no-empty-blocks
674         {} catch {
675             bytes memory reason = Exec.getReturnData(REVERT_REASON_MA
676             revert PostOpReverted(reason);
677         }
678     }
679     actualGas += preGas - gasleft();
680
681     // Calculating a penalty for unused execution gas
682     {
683         uint256 executionGasLimit = mUserOp.callGasLimit;
684         // Note that 'verificationGasLimit' here is the limit given to the 'p
685         if (context.length > 0){
686             executionGasLimit += mUserOp.verificationGasLimit;
687         }
688         uint256 executionGasUsed = actualGas - opInfo.preOpGas;
689         // this check is required for the gas used within EntryPoint and not
690         if (executionGasLimit > executionGasUsed) {
691             uint256 unusedGas = executionGasLimit - executionGasUsed;
692             uint256 unusedGasPenalty = (unusedGas * PENALTY_PERCENT) / 100;
693             actualGas += unusedGasPenalty;
694         }
695     }
696
697     actualGasCost = actualGas * gasPrice;
698     if (opInfo.prefund < actualGasCost) {
699         revert FailedOp(opIndex, "AA51 prefund below actualGasCost");
700     }
701     uint256 refund = opInfo.prefund - actualGasCost;
702     _incrementDeposit(refundAddress, refund);
703     bool success = mode == IPaymaster.PostOpMode.opSucceeded;
704     emit UserOperationEvent(
705         opInfo.userOpHash,
706         mUserOp.sender,
707         mUserOp.paymaster,
708         mUserOp.nonce,
709         success,
710         actualGasCost,
711         actualGas
712     );
713 } // unchecked
714
715 /**
716  * The gas price this UserOp agrees to pay.
717  * Relayer/block builder might submit the TX with higher priorityFee, but the use
718  * @param mUserOp - The userOp to get the gas price from.
719  */
720 function getUserOpGasPrice(
721     MemoryLocation memoryLocation

```

```

732         memoryuserop memory muserop
733     ) internal view returns (uint256) {
734         unchecked {
735             uint256 maxFeePerGas = mUserOp.maxFeePerGas;
736             uint256 maxPriorityFeePerGas = mUserOp.maxPriorityFeePerGas;
737             if (maxFeePerGas == maxPriorityFeePerGas) {
738                 //legacy mode (for networks that don't support basefee opcode)
739                 return maxFeePerGas;
740             }
741             return min(maxFeePerGas, maxPriorityFeePerGas + block.basefee);
742         }
743     }
744
745     /**
746      * The minimum of two numbers.
747      * @param a - First number.
748      * @param b - Second number.
749      */
750     function min(uint256 a, uint256 b) internal pure returns (uint256) {
751         return a < b ? a : b;
752     }
753
754     /**
755      * The offset of the given bytes in memory.
756      * @param data - The bytes to get the offset of.
757      */
758     function getOffsetOfMemoryBytes(
759         bytes memory data
760     ) internal pure returns (uint256 offset) {
761         assembly {
762             offset := data
763         }
764     }
765
766     /**
767      * The bytes in memory at the given offset.
768      * @param offset - The offset to get the bytes from.
769      */
770     function getMemoryBytesFromOffset(
771         uint256 offset
772     ) internal pure returns (bytes memory data) {
773         assembly {
774             data := offset
775         }
776     }
777
778     /// @inheritdoc IEntryPoint
779     function delegateAndRevert(address target, bytes calldata data) external {
780         (bool success, bytes memory ret) = target.delegatecall(data);
781         revert DelegateAndRevert(success, ret);
782     }
783 }

```

